@PACKAGE_INIT@

if(NOT TARGET tufao@TUFAO_VERSION_MAJOR@)
  include(${CMAKE_CURRENT_LIST_DIR}/Tufao@TUFAO_VERSION_MAJOR@Targets.cmake)
  set(Tufao@TUFAO_VERSION_MAJOR@_LIBRARY tufao@TUFAO_VERSION_MAJOR@)

  # For compatibility with previous find module
  set(TUFAO_VERSION_STRING "${Tufao@TUFAO_VERSION_MAJOR@_VERSION}")
  get_target_property(TUFAO_INCLUDE_DIR ${Tufao@TUFAO_VERSION_MAJOR@_LIBRARY} INTERFACE_INCLUDE_DIRECTORIES)
  get_target_property(_IMPORTED_CONFIGURATIONS ${Tufao@TUFAO_VERSION_MAJOR@_LIBRARY} IMPORTED_CONFIGURATIONS)
  list(GET _IMPORTED_CONFIGURATIONS 0 _IMPORTED_CONFIGURATION)
  string(TOUPPER "${CMAKE_BUILD_TYPE}" _CONFIG)
  foreach(c ${_IMPORTED_CONFIGURATIONS})
    if(c STREQUAL _CONFIG)
      set(_IMPORTED_CONFIGURATION "${c}")
    endif()
  endforeach()
  unset(_CONFIG)
  get_target_property(TUFAO_LIBRARIES ${Tufao@TUFAO_VERSION_MAJOR@_LIBRARY} IMPORTED_LOCATION_${_IMPORTED_CONFIGURATION})
  unset(_IMPORTED_CONFIGURATIONS)
  unset(_IMPORTED_CONFIGURATION)
endif()

